

On the Rosenberger Monster

Robert Fitzgerald Morse

ABSTRACT. In this paper we obtain structure results for the largest finite generalized triangle group that has been called the Rosenberger Monster. These structure results are motivated by their application for finding various homological functors for this group.

1. Introduction

A generalized triangle group is a group given by a presentation

$$\langle a, b \mid a^p = b^q = R^m = 1 \rangle,$$

where p, q, m are integers greater than 1, and R is a word of the form

$$a^{\alpha_1} b^{\beta_1} \dots a^{\alpha_k} b^{\beta_k},$$

with $1 \leq \alpha_i < p$, $1 \leq \beta_i < q$ for $i = 1, \dots, k$. It is assumed that m is chosen so that R is not a proper power. The largest finite generalized triangle group has been named the Rosenberger Monster. This group, which we denote as G throughout the paper, has the presentation

$$(1) \quad G = \langle a, b \mid a^2, b^3, (abababab^2ab^2abab^2ab^2)^2 \rangle.$$

The finite generalized triangle groups were classified by Howie, Metaftsis and Thomas [8], except for two possible groups, namely G and

$$H = \langle a, b \mid a^2, b^3, (abababab^2abab^2ab^2)^2 \rangle.$$

Lévai, Rosenberger and Souvignier [10] completed the classification by showing that G is finite and that H is an infinite group.

Lévai, Rosenberger and Souvignier state that using the computational group theory package GAP [7] one is able to show that G is finite of order $2^{20} \cdot 3^4 \cdot 5$. Moreover, G is a central product of $SL(2, 5)$ and a solvable group with amalgamated central subgroup of order 2. The subgroup structure of G was investigated by Howie, Metaftsis and Thomas in [9]. They identified the solvable group of the

2000 *Mathematics Subject Classification.* 20F05, 20J06, 20D99.

This paper is based in part on my talk at the AMS Special Session on Infinite Groups, Bard College, Annandale-on-Hudson, NY, October 9, 2005.

This research was supported by a grant from the Alumni Research and Scholarly Activity Fellowship (ARSAF) of the University of Evansville.

central product mentioned in [10] and computed Sylow subgroups of G for $p = 2$ and 3 based on this identification.

The purpose of this paper is to give further structure results for G and to show how these results can be applied in computing the nonabelian tensor square and nonabelian exterior square of G . From the nonabelian exterior square of G , one is able to compute several groups related to G , in particular, the second integral homology group, or Schur multiplier, of G . In Section 3 we show how our structure results can lead to finding these groups. In principle these computations can be done for the Rosenberger Monster, but are currently beyond the capability of the computing resources available to the author. The author hopes to complete these computations for all the finite generalized triangle groups for a future publication.

The proofs in this paper appeal to computations made in GAP. ¹ GAP commands and output are shown as part of the proofs. The reader can type in the GAP commands in sequential order as found in the text and obtain the results shown. Some of the computations take significant computer time and memory. All computations were done using a 3GHz processor with a GAP workspace of 2.8 gigabytes. ²

In this paper GAP commands will be written in a teletype font with all output suppressed using the double semicolon except where the output is of interest. A comment in GAP is any input following the # symbol. Some lines contain comments for clarity. Lines are broken and indented for readability. The GAP objects are named and constructed to represent mathematical objects in the text. As an example, we fix the GAP object G , which is a finitely presented group isomorphic to G , the Rosenberger Monster. It is constructed via the commands

```
gap> F := FreeGroup("a", "b");; a:=F.1;; b:=F.2;;
gap> R := [a^2, b^3,
>         (a*b*a*b*a*b*a*b^2*a*b^2*a*b*a*b^2*a*b^2)^2];;
gap> # Construct the finitely presented group
gap> G := F/R;
<fp group on the generators [ a, b ]>
gap> a:= G.1;; b:=G.2;; # Remember generators for later use
```

2. Structure results

We start by finding a subnormal series for G .

THEOREM 1. *Let G be the Rosenberger Monster. Then G has a subnormal series*

$$G \triangleright G^{(1)} \triangleright G^{(2)} \triangleright G^{(3)} \triangleright G^{(4)} \triangleright C \triangleright 1,$$

where $G^{(i)}$ is the i th term of the derived series and C is the center of $G^{(4)}$. The factors have the isomorphism types

$$\begin{aligned} G/G^{(1)} &\cong \mathbb{Z}_6, & G^{(1)}/G^{(2)} &\cong \mathbb{Z}_6 \times \mathbb{Z}_6, & G^{(2)}/G^{(3)} &\cong (\mathbb{Z}_2)^8, \\ G^{(3)}/G^{(4)} &\cong (\mathbb{Z}_2)^6, & G^{(4)}/C &\cong A_5 & \text{and } C &\cong \mathbb{Z}_2. \end{aligned}$$

PROOF. The following GAP commands can be used to directly find $G^{(i)}$ for $i = 1, 2, 3, 4$ and their abelian invariants.

¹The commands used in this paper require GAP version 4.4.7 (or higher). No GAP packages are required.

²The GAP workspace size was set using the `-m 2.8G` command line option.

```

gap> G1 := DerivedSubgroup(G);; G2 := DerivedSubgroup(G1);;
gap> G3 := DerivedSubgroup(G2);; G4 := DerivedSubgroup(G3);;
gap> AbelianInvariants(G); # Finds invariants for G/G'
[ 2, 3 ]
gap> AbelianInvariants(G1);
[ 2, 2, 3, 3 ]
gap> AbelianInvariants(G2);
[ 2, 2, 2, 2, 2, 2, 2, 2 ]
gap> AbelianInvariants(G3);
[ 2, 2, 2, 2, 2, 2 ]
gap> AbelianInvariants(G4);
[ ]

```

Our calculations above show that $G^{(4)}$ is perfect and $|G : G^{(4)}| = 2^{17} \cdot 3^3$. Since G has order $2^{20} \cdot 3^4 \cdot 5$, it follows that $G^{(4)}$ has order 120. There is only one perfect group of order 120, namely $SL(2, 5)$, and the result holds. \square

In the following proposition we give a complete description of $G^{(3)}$.

PROPOSITION 2. *Let G be the Rosenberger Monster. Then*

$$G^{(3)} \cong (\mathbb{Z}_2)^6 \times SL(2, 5).$$

PROOF. The GAP object **G3**, which is isomorphic to $G^{(3)}$, is not given as a finitely presented group. We first find a finitely presented group **G3fp** isomorphic to **G3** that is more efficient to compute with. We also name the isomorphism as it will be used in subsequent computations.

```

gap> G3fpiso := IsomorphismFpGroup(G3);; #Isomorphism G3 to fp group
gap> G3fp := Image(G3fpiso);
<fp group on the generators [ F1, F2, F3, F4, F5, F6 ]>
gap> Size(G3fp);
7680

```

To determine the structure of **G3fp**, we define a central subgroup **N** that is generated by the fifth powers of the generators of **G3fp**. We show this subgroup is isomorphic to $(\mathbb{Z}_2)^6$ and its intersection with the derived subgroup of **G3fp** is trivial.

```

gap> # Construct a subgroup N in the center of G3fp
gap> N := Subgroup(G3fp, List(GeneratorsOfGroup(G3fp), g->g^5));;
gap> IsSubgroup(Centre(G3fp), N);
true
gap> # See that N has the right structure
gap> StructureDescription(N);
"C2 x C2 x C2 x C2 x C2 x C2"
gap> D := DerivedSubgroup(G3fp);; #Isomormorphic to G4 = SL(2,5)
gap> Size(Intersection(N,D));
1

```

The normal subgroups **D** and **N** are of order 120 and 64 respectively, with trivial intersection. Hence $G^{(3)} \cong (\mathbb{Z}_2)^6 \times SL(2, 5)$. \square

We now define a set which plays a role in computing the nonabelian tensor square of a group.

DEFINITION 3. Let H be a group and let $H = H_n \supseteq \cdots \supseteq H_1 \supseteq H_0 = 1$ be a subnormal series for H . Let \mathcal{T}_i denote a transversal for H_{i-1} in H_i and let \mathcal{H}_i denote a lift of a generating set for H_{i-1}/H_i to \mathcal{T}_i . Set

$$\mathcal{L}_i = \begin{cases} \mathcal{H}_i, & \text{if } H_i/H_{i-1} \text{ is abelian} \\ \mathcal{T}_i, & \text{otherwise.} \end{cases}$$

Then define the set \mathcal{L}_H to be

$$\mathcal{L}_H = \cup_{i=1}^n \mathcal{L}_i.$$

The size of \mathcal{L}_H is related to the subnormal series one uses to define it. For any group H we can set $\mathcal{L}_H = H$ via $H \supseteq 1$. Finding a subnormal series of H with many abelian factors will reduce the size of \mathcal{L}_H .

In the following proposition we compute \mathcal{L}_G using the subnormal series found in Theorem 1. Conjugation on the right of two group elements x and y , as used in the following proposition, is defined as $x^y = y^{-1}xy$. Conjugation on the left, which is used in defining the nonabelian tensor square later, is defined as ${}^y x = yxy^{-1}$.

PROPOSITION 4. *Let G be the Rosenberger Monster. Then \mathcal{L}_G is the union of*

$$\begin{aligned} \mathcal{L}_{G/G^{(1)}} &= \{ab\}, \\ \mathcal{L}_{G^{(1)}/G^{(2)}} &= \{[a, b], [a, b^2]\}, \\ \mathcal{L}_{G^{(2)}/G^{(3)}} &= \{(ab)^6, (ab^2)^6, ((ab)^6)^b, ((ab)^6)^{ba}, ((ab)^6)^{bab}, ((ab)^6)^{bab^2}, \\ &\quad ((ab)^6)^{baba}, ((ab)^6)^{bab^2a}\}, \\ \mathcal{L}_{G^{(3)}/G^{(4)}} &= \{(ba)^{12}, (b^2a)^{12}, b \cdot (ab^2)^{11} \cdot ab, (babab)^6, [b^2, a]^6, [b, a]^6\}, \\ \mathcal{L}_{G^{(4)}/G} &= \{g_2, g_2^{-1}, g_2^2, g_2^{-2}, g_1, g_2^{-1}g_1, g_2g_1, g_2^2g_1, g_1^{-1}, g_2^{-1}g_1^{-1}, \\ &\quad g_2g_1^{-1}, g_2^{-2}g_1^{-1}, g_1^2, g_2^2g_1^2, g_2g_1^2, g_1^{-1}g_2^{-1}g_1, g_2^2g_1g_2, g_1g_2, \\ &\quad g_2g_1g_2, g_1g_2^{-1}, g_1g_2^{-1}g_1, g_1g_2^{-1}g_1^{-1}, g_1^{-2}, g_2^{-1}g_1^{-2}, g_2^{-2}g_1^{-2}, \\ &\quad g_1^{-1}g_2, g_1^{-1}g_2g_1, g_1^{-1}g_2^{-1}, g_2^{-1}g_1^{-1}g_2^{-1}, g_1g_2g_1^{-1}, g_1^2g_2, g_1^2g_2^{-1}, \\ &\quad g_1g_2^{-2}g_1^{-1}, g_1^{-1}g_2^2g_1g_2, g_1^{-2}g_2^2, g_1g_2g_1, g_2g_1g_2g_1, g_2g_1g_2^2, \\ &\quad g_1g_2^2, g_1g_2^{-2}, g_1^{-1}g_2^2g_1, g_1^{-2}g_2, g_1^{-2}g_2^{-1}, g_1g_2g_1g_2^2, g_1^{-1}g_2^2, \\ &\quad g_1^{-1}g_2^{-1}g_1^{-1}, g_2^{-1}g_1^{-1}g_2^{-1}g_1^{-1}, g_1^2g_2^{-2}, g_1^{-1}g_2^{-2}, g_1^2g_2g_1g_2, \\ &\quad g_1^2g_2g_1g_2^2, g_1^2g_2g_1, g_1^{-1}g_2^{-1}g_1^{-1}g_2^{-1}, g_1^2g_2^2, g_1g_2g_1^2, g_1^2g_2^{-2}g_1^{-1}, \\ &\quad g_1g_2g_1g_2g_1, g_1^{-2}g_2^{-2}, g_1g_2g_1g_2\}, \end{aligned}$$

where $g_1 = (ba)^{24}$, $g_2 = (b^2a)^{24}$, and

$$\mathcal{L}_{C/1_G} = \{(ab)^{24} \cdot (ab^2)^{48} \cdot (ab)^{24} \cdot (ab^2)^{96} \cdot (ab)^{96} \cdot (ab^2)^{96}\}.$$

PROOF. We construct the natural homomorphism $\phi : G \rightarrow G/G^{(1)}$ and show $\langle \phi(ab) \rangle = G/G^{(1)}$.

```
gap> nat := NaturalHomomorphismByNormalSubgroup(G, G1);;
gap> Subgroup(Image(nat), [Image(nat, a*b)])=Image(nat);
true
```

Similarly, we show the images of the elements of $\mathcal{L}_{G^{(i)}/G^{(i+1)}}$ under the natural homomorphism generate $G^{(i)}/G^{(i+1)}$ for $i = 1, 2$, as needed.

```
gap> nat := NaturalHomomorphismByNormalSubgroup(G1,G2);;
gap> h1 := Image(nat,Comm(a,b));; h2 := Image(nat,Comm(a,b^2));;
gap> Subgroup(Image(nat),[h1,h2])=Image(nat);
true

gap> nat := NaturalHomomorphismByNormalSubgroup(G2,G3);;
gap> h1 := Image(nat,(a*b)^6);; h2 := Image(nat,(a*b^2)^6);;
gap> h3 := Image(nat,((a*b)^6)^b);;
gap> h4 := Image(nat,((a*b)^6)^(b*a));;
gap> h5 := Image(nat,((a*b)^6)^(b*a*b));;
gap> h6 := Image(nat,((a*b)^6)^(b*a*b^2));;
gap> h7 := Image(nat,((a*b)^6)^(b*a*b*a));;
gap> h8 := Image(nat,((a*b)^6)^(b*a*b^2*a));;
gap> Subgroup(Image(nat),[h1,h2,h3,h4,h5,h6,h7,h8])=Image(nat);
true
```

To compute more efficiently, the remaining computations will be done in the finitely presented group $G3fp$ isomorphic to $G3$ by the mapping $G3fpiso$. We map the elements of $\mathcal{L}_{G^{(3)}/G^{(4)}}$ to $G3fp$ and show their images under the natural homomorphism generate $G3fp/G3fp' \cong G^{(3)}/G^{(4)}$.

```
gap> nat := NaturalHomomorphismByNormalSubgroup(
>   G3fp,DerivedSubgroup(G3fp));;
gap> h1 := Image(nat,Image(G3fpiso,(b*a)^12));;
gap> h2 := Image(nat,Image(G3fpiso,(b^2*a)^12));;
gap> h3 := Image(nat,Image(G3fpiso,b*(a*b^2)^11*a*b));;
gap> h4 := Image(nat,Image(G3fpiso,(b*a*b*a*b)^6));;
gap> h5 := Image(nat,Image(G3fpiso,Comm(b^2,a)^6));;
gap> h6 := Image(nat,Image(G3fpiso,Comm(b,a)^6));;
gap> Subgroup(Image(nat),[h1,h2,h3,h4,h5,h6])=Image(nat);
true
```

To find the transversal for $G^{(4)}/C$ as listed in the proposition, we need to set up in GAP

$$\langle g_1, g_2 \rangle = G^{(4)} \cong SL(2, 5),$$

where $g_1 = (ba)^{24}$ and $g_2 = (b^2a)^{24}$. We first map the elements $(b*a)^{24}$ and $(b^2*a)^{24}$, which are elements in $G3$, to $G3fp$, and form a subgroup of $G3fp$ generated by their images. We then compute an isomorphism by generators to a finitely presented group with generators $g1$ and $g2$ and find the required transversal.

```
gap> g1 := (b*a)^24;; g2 := (b^2*a)^24;;
gap> # Find subgroup generated by the images g1 and g2 in G3fp
gap> # and compute a finite presentation preserving the generators
gap> iso := IsomorphismFpGroupByGenerators(
>   Subgroup(G3fp,[Image(G3fpiso,g1),Image(G3fpiso,g2)]),
>   [Image(G3fpiso,g1),Image(G3fpiso,g2)],"g");;
gap> # Check we have the right group
gap> StructureDescription(Image(iso));
```

```
"SL(2,5)"
gap> # List the transversal elements as given in the proposition
gap> Elements(RightTransversal(Image(iso),Centre(Image(iso))));;
```

We suppressed the output of the last command as it mimics verbatim the list of elements found in the proposition. The following is a sample check to demonstrate that the correspondence $g_1 = (ba)^{24}$ and $g_2 = (b^2a)^{24}$ is maintained in our GAP calculations, as claimed.

```
gap> # Compute the right cosets of G4/C in our isomorphic image of G4
gap> rcs := RightCosets(Image(iso),Centre(Image(iso)));;
gap> # Check (b*a)^24*((b^2*a)^24)^4*(b*a)^24 maps to the
gap> # correct coset
gap> Representative(First(rcs,
> rc->Image(iso,
> Image(G3fpiso, (b*a)^24*((b^2*a)^24)^4*(b*a)^24)) in rc));
g1*g2^4*g1
```

Finally, we show that $(ab)^{24}(ab^2)^{48}(ab)^{24}(ab^2)^{96}(ab)^{96}(ab^2)^{96}$ generates the center of $G^{(4)}$, as needed.

```
gap> c := (a*b)^24*(a*b^2)^48*(a*b)^24*(a*b^2)^96*(a*b)^96*
> (a*b^2)^96;;
gap> Subgroup(G3fp, [Image(G3fpiso, c)])=
> Centre(DerivedSubgroup(G3fp));
true
```

□

We conclude this section by finding the orders of three elements used in defining \mathcal{L}_G . From Proposition 4 we see that $(ab)^{12}$, $[a, b]^6$ and $[a, b^2]^6$ are elements of $G^{(3)}$. We can compute their orders in G3fp as follows

```
gap> Order(Image(G3fpiso, (a*b)^12));
10
gap> Order(Image(G3fpiso, Comm(a, b)^6));
10
gap> Order(Image(G3fpiso, Comm(a, b^2)^6));
10
```

Hence $|ab| = 120$ and $|[a, b]| = |[a, b^2]| = 60$.

3. Computing homological functors for G

The motivation for investigating a subnormal series for G in Section 2 is to aid in the computation of the various groups related to G found in the following

commutative diagram

$$\begin{array}{ccccccc}
 & & & 0 & & 0 & \\
 & & & \downarrow & & \downarrow & \\
 H_3(G) & \longrightarrow & \Gamma(G^{ab}) & \xrightarrow{\psi} & J_2(G) & \longrightarrow & H_2(G) \longrightarrow 0 \\
 \parallel & & \parallel & & \downarrow & & \downarrow \\
 (2) \quad H_3(G) & \longrightarrow & \Gamma(G^{ab}) & \xrightarrow{\psi} & G \otimes G & \longrightarrow & G \wedge G \longrightarrow 1 \\
 & & & & \kappa \downarrow & & \kappa' \downarrow \\
 & & & & G' & \xlongequal{\quad} & G' \\
 & & & & \downarrow & & \downarrow \\
 & & & & 1 & & 1
 \end{array}$$

with exact rows and central extensions as columns [5]. The group $\Gamma(G^{ab})$ is Whitehead's quadratic functor (see [5]), $J_2(G)$ is isomorphic to the third homotopy group of the suspension of an Eilenberg-MacLane space of G , $H_2(G)$ and $H_3(G)$ are the second and third integral homology groups of G , and $G \otimes G$ and $G \wedge G$ are the nonabelian tensor square and nonabelian exterior square of G , respectively.

The nonabelian tensor square of a group H is the group generated by the symbols $g \otimes h$ for all g, h in H subject to the relations

$$gg' \otimes h = ({}^g g' \otimes {}^g h)(g \otimes h) \quad \text{and} \quad g \otimes hh' = (g \otimes h)({}^h g \otimes {}^h h')$$

for all g, g', h, h' in H . The subgroup $\nabla(H)$ of $H \otimes H$ is the central subgroup generated by the elements of $h \otimes h$ for all $h \in H$. The nonabelian exterior square of H is defined as $H \wedge H = H \otimes H / \nabla(H)$.

Finding the nonabelian tensor square is an important step in finding the other groups in the commutative diagram. Using the definition above to compute the nonabelian tensor square of the Rosenberger Monster would involve simplifying a group presentation with $2^{40} \cdot 3^8 \cdot 5^2$ generators and $2^{61} \cdot 3^{12} \cdot 5^3$ relations, which is intractable by direct calculation. Another approach for computing the nonabelian tensor square is the "crossed pairing" method, which has been used successfully for computing the nonabelian tensor square of infinite groups. For some examples, see [1], [2], [3] and [4]. The main drawback of this method for our application is that the tensor square essentially needs to be known a priori. The crossed pairing is set up with this "advance" knowledge, from which a proof that the group in question really is the nonabelian tensor square can be completed.

Another approach involves a group construction due to Rocco [12]. Let H and H^φ be isomorphic groups via $\varphi : h \rightarrow h^\varphi$. Define the group $\nu(H)$ by

$$\nu(H) = \langle H, H^\varphi \mid {}^k [g, h^\varphi] = [{}^k g, ({}^k h)^\varphi] = {}^{k^\varphi} [g, h^\varphi] \text{ for all } g, h, k \in H \rangle.$$

The group $\nu(H)$ can also be found in the work of Ellis [6]. It is related to $H \otimes H$ in the following way.

THEOREM 5 ([6],[12]). *Let H be a group. The subgroup $[H, H^\varphi]$ of $\nu(H)$ is isomorphic to $H \otimes H$.*

The following theorem gives a different presentation of $\nu(H)$ that involves \mathcal{L}_H .

THEOREM 6 ([11]). *Let H be a group generated by a set \mathcal{H} . Then $\nu(H) = H * H^\varphi / \langle J \rangle$, where J is the normal generating set consisting of the elements*

$${}^x[a, b^\varphi][{}^x a, ({}^x b)^\varphi]^{-1}, {}^{x^\varphi}[a, b^\varphi][{}^x a, ({}^x b)^\varphi]^{-1}$$

for all a, b in \mathcal{H} and x in \mathcal{L}_H .

The presentation of $\nu(H)$ can be greatly simplified whenever H has a subnormal series with many abelian factors. As noted in Section 2, this minimizes \mathcal{L}_H . For G , the Rosenberger Monster, we can determine a presentation for $\nu(G)$ from the results found in Section 2.

PROPOSITION 7. *Let G be the Rosenberger Monster with presentation (1). Then $\nu(G)$ is the finitely presented group generated by a, b, a^φ and b^φ with relations*

$$\begin{aligned} a^2, \quad b^3, \quad (abababab^2ab^2abab^2ab^2)^2, \\ (a^\varphi)^2, \quad (b^\varphi)^3, \quad ((abababab^2ab^2abab^2ab^2)^\varphi)^2, \\ {}^x[a, a^\varphi] = [{}^x a, ({}^x a)^\varphi] = {}^{x^\varphi}[a, a^\varphi], \quad {}^x[a, b^\varphi] = [{}^x a, ({}^x b)^\varphi] = {}^{x^\varphi}[a, b^\varphi], \\ {}^x[b, a^\varphi] = [{}^x b, ({}^x a)^\varphi] = {}^{x^\varphi}[b, a^\varphi] \quad \text{and} \quad {}^x[b, b^\varphi] = [{}^x b, ({}^x b)^\varphi] = {}^{x^\varphi}[b, b^\varphi] \end{aligned}$$

for all x in \mathcal{L}_G .

To actually compute the subgroup $[G, G^\varphi]$ for $\nu(G)$ as presented in Proposition 7 seems difficult. The group $\nu(G)$ is not solvable since it contains an isomorphic copy of G . We only know that $\nu(G)$ is finite [12].

It was stated in [10] and [9] that G is the central product of a solvable group A and $SL(2, 5)$, where the center of $SL(2, 5)$ is amalgamated with a subgroup of the center of A . Specifically, G is equal to a quotient of $A \times SL(2, 5)$ by a cyclic subgroup of order 2. Since G is a homomorphic image of $A \times SL(2, 5)$, it follows that the nonabelian tensor square of G is a homomorphic image of

$$(A \times SL(2, 5)) \otimes (A \times SL(2, 5))$$

(Proposition 1, [5]). By Proposition 11 of [5], we have

$$\begin{aligned} (A \times SL(2, 5)) \otimes (A \times SL(2, 5)) \\ = (A \otimes A) \times (A \otimes SL(2, 5)) \times (SL(2, 5) \otimes A) \times (SL(2, 5) \otimes SL(2, 5)), \end{aligned}$$

where $SL(2, 5) \otimes A$ and $A \otimes SL(2, 5)$ are ordinary tensor products.

Since $SL(2, 5)$ is perfect, the Schur cover of $SL(2, 5)$ is unique and equal to $SL(2, 5) \otimes SL(2, 5)$ (Corollary 1, [5]). Hence, using the following GAP commands

```
gap> SC := SchurCover(SL(2,5)); ;
gap> StructureDescription(SC);
"SL(2,5)"
```

shows $SL(2, 5) \otimes SL(2, 5) = SL(2, 5)$.

So finding the tensor square of $A \times SL(2, 5)$ reduces to finding $A \otimes A$. Since A is solvable, $\nu(A)$ is also solvable [12]. The question is whether we are able to find a polycyclic presentation for $\nu(A)$. Once such a presentation is found, it is easy to compute $[A, A^\varphi]$. The author was unable to find a polycyclic presentation for $\nu(A)$ using the Solvable Quotient (SQ) algorithm found in GAP or using an experimental polycyclic quotient algorithm. The computational tractability of finding a polycyclic presentation is in part related to the presentation the algorithms are given. Finding $A \otimes A$ is possible but it may take a combination of hand calculations

and computer assisted work. These hand calculations use a commutator calculus related to the subgroup $[A, A^\varphi]$. A full treatment of these techniques will be the topic of a future publication.

We conclude with a description of A and \hat{A} , the largest solvable subgroup in G .

PROPOSITION 8. *Let G be the Rosenberger Monster with presentation (1). The subgroups*

$$\hat{A} = \langle a, (b^a)^b, (b^a)^{b^2} \rangle \text{ and } A = \langle a, b \cdot [a, b^2], b^2 \cdot [a, b^2] \cdot [a, b]^2, b^2 \cdot (ab)^{12} \rangle$$

are solvable subgroups of G of index 5 and 60, respectively. Moreover, the group \hat{A} is the largest solvable subgroup of G and G is the central product of A and $G^{(4)}$.

PROOF. We first show that \hat{A} and A have the proper indices in G .

```
gap> Ahat := Subgroup(G, [a, (b^a)^b, (b^a)^(b^2)]);;
gap> Index(G,Ahat);
5
gap> A := Subgroup(G, [a, b*Comm(a, b^2), b^2*Comm(a, b^2)*Comm(a, b)^2,
> b^2*(a*b)^12]);;
gap> Index(G,A);
60
```

It follows from the order of G that $|\hat{A}| = 2^{20} \cdot 3^4$ and $|A| = 2^{18} \cdot 3^3$. We compute a solvable quotient for both \hat{A} and A of these exact orders.

```
gap> # Obtain a finite presentation for Ahat and A
gap> Ahatfp := Image(IsomorphismFpGroup(Ahat));;
gap> Afp := Image(IsomorphismFpGroup(A));;
gap> # Show these groups are solvable by finding a solvable
gap> # quotient of each group with the same order
gap> Ahatpc := Image(
> EpimorphismSolvableQuotient(Ahatfp, 2^20*3^4));;
gap> Collected(Factors(Size(Ahatpc)));
[[ 2, 20 ], [ 3, 4 ] ]
gap> Apc := Image(EpimorphismSolvableQuotient(Afp, 2^18*3^3));;
gap> Collected(Factors(Size(Apc)));
[[ 2, 18 ], [ 3, 3 ] ]
```

Hence \hat{A} and A are solvable subgroups, as needed.

To see that \hat{A} is the largest solvable subgroup of G , we show that any other subgroup with smaller index contains the nonsolvable subgroup $G^{(3)}$.

```
gap> # Find all subgroups with index 4 or less.
gap> lws := Filtered(LowIndexSubgroupsFpGroup(G, 4),
> x->Index(G,x) in [2,3,4]);;
gap> ForAll(lws, x->IsSubgroup(x, G3));
true
```

In Proposition 4 we showed that $\langle (ba)^{24}, (b^2a)^{24} \rangle = G^{(4)}$ and we identified the generator of C as the GAP variable c . We show that $A \cap G^{(4)} = C$ and that $\langle A, G^{(4)} \rangle = G$, as needed.

```

gap> # c is in A
gap> c in A;
true
gap> # Show the subgroup generated by A and G4 is all of G
gap> Index(G, Subgroup(G, Concatenation(GeneratorsOfGroup(A),
>                                     [(b*a)^24, (b^2*a)^24]));
1
gap> I := Intersection(A, G3);;
gap> GeneratorsOfGroup(I);;
gap> S := Subgroup(G3fp, List(GeneratorsOfGroup(I),
>                             x->Image(G3fpiso, x)));;
gap> Ifp := Intersection(S, DerivedSubgroup(G3fp));;
gap> Size(Ifp);
2
gap> Ifp = Centre(DerivedSubgroup(G3fp));
true

```

□

Acknowledgments. The author would like to thank Luise-Charlotte Kappe and Russell Blyth for their helpful comments and proofreading the final version of the paper.

References

- [1] Michael R. Bacon. On the nonabelian tensor square of a nilpotent group of class two. *Glasgow Math. J.*, 36(3):291–296, 1994.
- [2] Michael R. Bacon, Luise-Charlotte Kappe, and Robert Fitzgerald Morse. On the nonabelian tensor square of a 2-Engel group. *Arch. Math. (Basel)*, 69(5):353–364, 1997.
- [3] James R. Beuerle and Luise-Charlotte Kappe. Infinite metacyclic groups and their non-abelian tensor squares. *Proc. Edinburgh Math. Soc. (2)*, 43(3):651–662, 2000.
- [4] Russell D. Blyth, Robert Fitzgerald Morse, and Joanne L. Redden. On computing the non-abelian tensor squares of the free 2-Engel groups. *Proc. Edinb. Math. Soc. (2)*, 47(2):305–323, 2004.
- [5] R. Brown, D. L. Johnson, and E. F. Robertson. Some computations of nonabelian tensor products of groups. *J. Algebra*, 111(1):177–202, 1987.
- [6] Graham Ellis and Frank Leonard. Computing Schur multipliers and tensor products of finite groups. *Proc. Roy. Irish Acad. Sect. A*, 95(2):137–147, 1995.
- [7] The GAP Group. *GAP – Groups, Algorithms, and Programming, Version 4.4*, 2005. (<http://www.gap-system.org>).
- [8] J. Howie, V. Metaftsis, and R. M. Thomas. Finite generalized triangle groups. *Trans. Amer. Math. Soc.*, 347(9):3613–3623, 1995.
- [9] James Howie, Vasileios Metaftsis, and Richard M. Thomas. Triangle groups and their generalisations. In *Groups—Korea '94 (Pusan)*, pages 135–147. de Gruyter, Berlin, 1995.
- [10] L. Lévai, G. Rosenberger, and B. Souvignier. All finite generalized triangle groups. *Trans. Amer. Math. Soc.*, 347(9):3625–3627, 1995.
- [11] Aidan McDermott. *Tensor Products of Groups and Related Theory*. PhD thesis, National University of Ireland Galway, May 1998.
- [12] N. R. Rocco. On a construction related to the nonabelian tensor square of a group. *Bol. Soc. Brasil. Mat. (N.S.)*, 22(1):63–79, 1991.

DEPARTMENT OF ELECTRICAL ENGINEERING AND COMPUTER SCIENCE, UNIVERSITY OF EVANSVILLE,
EVANSVILLE IN 47722 USA

E-mail address: rfmorse@evansville.edu

URL: faculty.evansville.edu/rm43